EE230 - Experiment 3

Param Rathour - 190070049
Autumn Semester 2021-22

1 Common-Emitter Amplifier

1.1 Biasing Circuit

1.1.1 Analysis

First, apply Thevenin's tTeorem at the base terminal

$$V_{\text{th}} = V_{CC} \cdot \left(\frac{R_2}{R_1 + R_2}\right) = \frac{132}{61} \quad R_{\text{th}} = \frac{R_1 \cdot R_2}{R_1 + R_2} = \frac{110000}{61}$$

As $V_{\text{th}} > 0.7$, the device is not in cutoff mode. Now by KVL,

$$V_{\text{th}} - V_{BE} = I_B \cdot R_{\text{th}} + I_E \cdot R_E$$

$V_{BE} = 0.7V$ and $I_E = (\beta + 1)I_B$ (where $\beta = 200$) gives,

$$I_B = \frac{V_{\text{th}} - V_{BE}}{R_{\text{th}} + (\beta + 1)R_E} = \frac{\frac{132}{61} - 0.7}{\frac{110000}{61} + (200 + 1) \cdot 1000} = 7.2185 \cdot 10^{-6} \quad \text{So, } I_C = \beta \cdot I_B = 1.4437 \cdot 10^{-3}$$

$$V_C = V_{CC} - I_C \cdot R_C = 12 - 1.4437 \cdot 10^{-3} \cdot 1200 = 10.267 \quad V_E = I_E \cdot R_E = 1.4509 \cdot 10^{-3} \cdot 1000 = 1.4509$$

As $V_C > 0$, the device is in **active** mode.
$I_B = 7.2185 \cdot 10^{-6}A, \quad I_C = 1.4437 \cdot 10^{-3}A, \quad V_B = 2.1509V, \quad V_C = 10.267V, \quad V_E = 1.4509V$

1.1.2 NGSPICE Values

$I_B = 1.125117 \cdot 10^{-5}A, \quad I_C = 1.462957 \cdot 10^{-3}A, \quad V_B = 2.143645V, \quad V_C = 10.24445V, \quad V_E = 1.474208V$

1.1.3 Code

```
Param Rathour (190070049), Biasing Circuit

.include bc547.txt                                  ; Includes BJT Model
Q1 c b e bc547a
VCC in gnd 12                                       ; Supply Voltage
Vdc d1 c 0                                          ; Dummy Voltage for IB
Vdb mid b 0                                         ; Dummy Voltage for IC

RC in d1 1.2k                                       ; Resistor
R1 in mid 10k                                       ; Resistor
R2 mid gnd 2.2k                                     ; Resistor
RE e gnd 1k                                         ; Resistor
CE e gnd 100u                                       ; Capacitor

.op                                                 ; Operating Point Analysis
.control                                            ; Control Functions
run
print v(c) v(b) v(e)
print i(Vdc) i(Vdb)
.endc
.end
```

1.1.4 Learnings

Learned to calculate operating point and appreciate Thevenin's theorem
This amplifier faces problems when R_{in} is low and/or R_{out} is high as it has medium R_{in} & R_{out}
I_B NGSPCIE values are not matching with analysis, this could be because of different BJT model parameters

1.2 Midband Voltage Gain and Frequency Response (with bypass Capacitor C_E)

1.2.1 NGSPICE Values

Midband Voltage Gain $= 3.47490 \cdot 10^1$dB, $f_L = 8.31492 \cdot 10^1$Hz, $f_H = 2.71684 \cdot 10^6$Hz, Bandwidth $= 2.71676 \cdot 10^6$Hz

1.2.2 Plot



(a) V_{out} vs V_{in}



(b) V_{out} vs V_{in} (in dB)



1.2.3 Code

```
Param Rathour (190070049), Midband Voltage Gain and Frequency Response

.include bc547.txt                                    ; Includes BJT Model
Q1 c b e bc547a
Vin in gnd dc 0 ac 10m                                ; Input Voltage
VCC Vcc gnd 12                                        ; Supply Voltage

RC Vcc c 1.2k                                         ; Resistor
R1 Vcc b 10k                                          ; Resistor
R2 b gnd 2.2k                                         ; Resistor
RL out gnd 100k                                       ; Resistor
RE e gnd 1k                                           ; Resistor
RS Vs in 0                                            ; Resistor
CE e gnd 100u                                         ; Capacitor
C1 b Vs 10u                                           ; Capacitor
C2 c out 10u                                          ; Capacitor

.ac DEC 10 1 500000k                                  ; AC Analysis
.control                                              ; Control Functions
run
meas ac Voutmax max vdb(out)
meas ac Vinmag max vdb(in)
let Vdbreq = Voutmax-3
meas ac fL when vdb(out) = Vdbreq rise = 1
meas ac fH when vdb(out) = Vdbreq fall = 1

let midbandVoltageGain = Voutmax-Vinmag
let bandwidth = fH - fL
plot vdb(in) vdb(out) xlog
print midbandVoltageGain fL fH bandwidth
.endc
.end
```

1.2.4 Learnings

High voltage gain and fairly large 3dB Bandwidth for Midband Voltage Gain in Common-Emitter Amplifier.

1.3 Effect of R_L, R_S on the Midband Gain

Table 1.1: Effect of R_L

R_S	R_L (in $k\Omega$)	Midband Voltage Gain (in dB)
0	1.2	28.91650
0	12	34.03839
0	120	34.76586
0	1200	34.84207

Table 1.2: Effect of R_S

R_S (in $k\Omega$)	R_L (in $k\Omega$)	Midband Voltage Gain (in dB)
2.2	100	25.28289
10	100	14.77365
20	100	9.199860
50	100	1.520630

Table 1.3: Common-Emitter Amplifier (with bypass Capacitor CE)

1.3.1 Plots



(a) $R_L = 1.2k\Omega$ & $R_S = 0\Omega$



(b) $R_L = 12k\Omega$ & $R_S = 0\Omega$



(c) $R_L = 100k\Omega$ & $R_S = 2.2k\Omega$



(d) $R_L = 100k\Omega$ & $R_S = 10k\Omega$

1.3.2 Code

```
Param Rathour (190070049), Effect of RL & RS on the Midband Gain
.include bc547.txt                                  ; Includes BJT Model
Q1 c b e bc547a
Vin in gnd dc 0 ac 10m                              ; Input Voltage
VCC Vcc gnd 12                                      ; Supply Voltage

RC Vcc c 1.2k                                       ; Resistor
R1 Vcc b 10k                                        ; Resistor
R2 b gnd 2.2k                                       ; Resistor
RL out gnd 1.2k                                     ; Resistor
RE e gnd 1k                                         ; Resistor
RS Vs in 0                                          ; Resistor
CE e gnd 100u                                       ; Capacitor
C1 b Vs 10u                                         ; Capacitor
C2 c out 10u                                        ; Capacitor

.ac DEC 10 1 500000k                                ; AC Analysis
.control                                            ; Control Functions
run
meas ac Voutmax max vdb(out)
meas ac Vinmag max vdb(in)
let midbandVoltageGain = Voutmax - Vinmag
plot vdb(in) vdb(out) xlog
print midbandVoltageGain
.endc
.end
```

1.3.3 Learnings

For a particular R_S, Midband Voltage Gain increases with increase in R_L and approaches a constant at high R_L.
For a particular R_L, Midband Voltage Gain decreases with increase in R_S and becomes negative at high R_S values.

2 Two-Stage Amplifier

2.1 Biasing Circuit

2.1.1 NGSPICE Values

$I_{B_2} = 8.075083 \cdot 10^{-6}A, \quad I_{C_2} = 9.576092 \cdot 10^{-4}A, \quad V_{E_2} = 9.576092V$

2.2 Midband Voltage Gain and Frequency Response

2.2.1 NGSPICE Values

Midband Voltage Gain $= 3.47767 \cdot 10^{1}$dB, Bandwidth $= 2.52560 \cdot 10^{6}$Hz, $f_L = 8.31666 \cdot 10^{1}$Hz, $f_H = 2.52568 \cdot 10^{6}$Hz

2.2.2 Plots



(a) V_{out} vs V_{in}



(b) V_{out} vs V_{in} (in dB)



2.2.3 Code

```
Param Rathour (190070049), Biasing Circuit, Midband Voltage Gain and Frequency Response
.include bc547.txt                                   ; Includes BJT Model
Q1 c1 b1 e1 bc547a
Q2 VCC b2 e2 bc547a
Vin in gnd dc 0 ac 1                                 ; Input Voltage
VCC Vcc gnd 12                                       ; Supply Voltage
Vdb2 c1 b2 0                                         ; Dummy Voltage
Vde2 e2 d1 0                                         ; Dummy Voltage
RC Vcc c1 1.2k                                       ; Resistor
R1 Vcc b1 10k                                        ; Resistor
R2 b1 gnd 2.2k                                       ; Resistor
RL out gnd 10k                                       ; Resistor
RE e1 gnd 1k                                         ; Resistor
RE2 d1 gnd 10k                                       ; Resistor
RS Vs in 0                                           ; Resistor
C1 b1 Vs 10u                                         ; Capacitor
C2 e2 out 10u                                        ; Capacitor
CE e1 gnd 100u                                       ; Capacitor
.ac DEC 10 1 500000k                                 ; AC Analysis (.op for biasing)
.control                                             ; Control Functions
run
meas ac Voutmax max vdb(out)
meas ac Vinmag max vdb(in)
let Vdbreq = Voutmax-3
meas ac fL when vdb(out) = Vdbreq rise = 1
meas ac fH when vdb(out) = Vdbreq fall = 1
let midbandVoltageGain = Voutmax-Vinmag
let bandwidth = fH - fL
print V(e2) i(Vdb2) i(Vde2)                          ; Use when using .op
plot vdb(in) vdb(out) xlog
print midbandVoltageGain fL fH bandwidth
```

2.2.4 Learnings

Midband Voltage Gain increases with increase in R_L. This gain approaches a constant at higher R_L values.