EE236: Experiment 3
I/V and C/V Characteristics of Solar Cell

Param Rathour, 190070049
Spring Semester, 2021-22

Contents

1 Overview of the experiment

This report contains my approach to the experiment, the circuit's design with the relevant simulation code and output plots.

1.1 Aim of the experiment

- To plot and understand I/V characteristics of solar cell under different lighting conditions and calculate fill factor
- To plot and understand C/V characteristics of solar cell and calculate built-in potential and doping concentration

1.2 Methods

First a comparison between different lighting conditions was made using the parameters (ideality factor) from the I/V characteristics of solar cell.
Then, C/V characteristics of a Solar Cell was calculated using the given circuit made of inverting summing amplifier and current to voltage converter and the parameters were calculated appropriately.

2 Design

I/V characteristics circuit was taken from experiment 3 handout (Figure 1a).
Voltage was varied from -2 to $+2$.



(a) For I/V characteristics



(b) Measurement of V_{OC} and I_{SC}

Analytically,

$$I = I_0\left(e^{\frac{q\cdot V_D}{\eta kT}} - 1\right) - I_L \tag{1}$$

By taking natural logarithm of I and neglecting -1, we get $\log(I + I_L)/V$ characteristics

$$\ln\left(I + I_L\right) = \ln(I_0) + \frac{q\cdot V}{\eta kT} \quad\rightarrow\quad \ln(I + I_L) = \underbrace{\frac{q}{\eta kT}}_{slope}\cdot V + \underbrace{\ln(I_0)}_{y-intercept} \tag{2}$$

This graph is a straight line in some range of I, V. We can calculate its slope to get η and y-intercept by interpolating that straight line to get I_S

$$\text{slope} = \frac{\ln(I + I_L)_2 - \ln(I + I_L)_1}{V_2 - V_1} = \frac{1}{\eta V_T} \rightarrow \eta = \frac{1}{V_T}\left(\frac{V_{D_2} - V_{D_1}}{\ln(I + I_L)_2 - \ln(I + I_L)_1}\right) \tag{3}$$

Then, the circuit for measurement of V_{OC} and I_{SC} was also taken from slides (Figure 1b)
Here, all parameters were calculated from NGSPICE commands itself.
I_{SC} is maximum current through the cell.
V_{OC} is maximum voltage across the cell.
I_{MP}, V_{MP} and P_{MP} were taken from $P_R - V_D$ graph.
Fill Factor $= \dfrac{I_{\text{MP}} \cdot V_{\text{MP}}}{I_{\text{SC}} \cdot V_{\text{OC}}}$
The circuit for C/V measurement is taken from slides too (Figure 2),
Voltage dc1 is varied from 0 to 2V and V_{sin} is 1V peak to peak 1kHz.



Figure 2: C/V Measurement

In C_{DUT} vs V_{dc}, C is calculated using V_{out} by below equation

$$\left|\frac{V_{\text{out}}}{V_{\text{DUT}}}\right| = \frac{C_{\text{DUT}}}{C_{\text{fb}}} \cdot \frac{1}{\sqrt{1 + \frac{1}{(\omega R_{\text{fb}} \cdot C_{\text{fb}})^2}}}$$

So,

$$C_{\text{DUT}} = \left|\frac{V_{\text{out}}}{V_{\text{DUT}}}\right| \cdot C_{\text{fb}} \cdot \sqrt{1 + \frac{1}{(\omega R_{\text{fb}} \cdot C_{\text{fb}})^2}} \tag{4}$$

C is normalised capacitance $= \frac{C_{\text{DUT}}}{\text{Area}}$, where area $= 4cm \cdot 4cm$
Built-in potential and doping concentration were calculated using $1/C^2$ vs V_{DUT} plot and the following equation

$$\frac{1}{C^2} = \underbrace{\frac{2}{q\varepsilon N_D} \cdot V_{\text{bi}}}_{\text{y-intercept}} - \underbrace{\frac{2}{q\varepsilon N_D}}_{\text{slope}} \cdot V_{\text{R}} \tag{5}$$

3 Simulation results

3.1 Plots

3.1.1 I/V Characteristics of Solar

Ideality factor for the three cases are
$I_L = 0$, $\eta = 3.44781$
$I_L = 8mA$, $\eta = 4.026876$
$I_L = 10mA$, $\eta = 4.151296$



(a) I/V when $I_L = 0$



(b) $\log I/V$ when $I_L = 0$



(c) I/V when $I_L = 8mA$



(d) $\log I/V$ when $I_L = 8mA$



(e) I/V when $I_L = 10mA$



(f) $\log I/V$ when $I_L = 10mA$

Figure 3: I/V Characteristics of Solar Cell

3.1.2 Measurement of V_{OC} and I_{SC} and fill factor of solar cell

I_L(in mA)	I_{SC}(in mA)	V_{OC}(in mV)	V_{MP}(in mV)	I_{MP}(in mA)	Fill Factor
8	7.90431	412.0659	276.429	6.19565	0.5312704
10	9.87702	424.9854	275	7.78378	0.5116530

Table 1: Band Gap for different diodes using Peak Emission Wavelength



(a) $I_D - V_D$ when $I_L = 8mA$



(b) $P_R - V_D$ when $I_L = 8mA$



(c) $I_D - V_D$ when $I_L = 10mA$



(d) $P_R - V_D$ when $I_L = 10mA$

Figure 4: Measurement of V_{OC} and I_{SC} and factor of solar cell

3.1.3 Solar Cell C/V Measurement

$N_D = 4.475974 \cdot 10^{14}$ atoms$\cdot cm^{-3}$ $V_{bi} = 0.782769V$



(a) C_{DUT} vs V_{dc}

(b) $1/C^2$ vs V_{DUT}

Figure 5: Solar Cell C/V Measurement

3.2 Code Snippets

3.2.1 I/V Characteristics of Solar

3.2.1.1 $I_L = 0$

```
Param Rathour (190070049), I/V characteristics of Solar Cell

.include Solar_Cell.txt                 ; Includes Solar Cell Model

R1 mid out 100
X1 out gnd solar_cell IL_val = 0        ; Solar Cell
Vin in gnd dc 0                         ; DC source Vin
Vdummy in mid 0                         ;
.dc Vin -2 2 0.01                       ; DC Analysis
.control
run
plot I(Vdummy) vs V(out)
plot log(abs(I(Vdummy))) vs V(out)
let VD1 = 0.15
let VD2 = 0.4
meas dc ID1 FIND I(Vdummy) WHEN V(out) = VD1
meas dc ID2 FIND I(Vdummy) WHEN V(out) = VD2
```

```
let lnID1 = log(ID1)
let lnID2 = log(ID2)
let slope = (lnID2 - lnID1) / (VD2 - VD1)
* let yintercept = lnID1 - slope * VD1
let eta = (VD2 - VD1) / ((26/1000) * (lnID2 - lnID1))
print eta                               ; , exp(yintercept)

hardcopy 1a1.eps I(Vdummy) vs V(out)
hardcopy 1a2.eps log(abs(I(Vdummy))) vs V(out)
.endc
.end
```

3.2.1.2 $I_L = 8mA$

```
Param Rathour (190070049), I/V characteristics of Solar Cell

.include Solar_Cell.txt                 ; Includes Solar Cell Model

R1 mid out 100
X1 out gnd solar_cell IL_val = 8e-3     ; Solar Cell
Vin in gnd dc 0                         ; DC source Vin
Vdummy in mid 0                         ;
.dc Vin -2 2 0.01                       ; DC Analysis
.control
run
plot I(Vdummy) vs V(out)
let y = log({abs(I(Vdummy)+8m)})
plot y vs V(out)
let VD1 = 0
let VD2 = 0.55
meas dc lnID1 FIND y WHEN V(out) = VD1
meas dc lnID2 FIND y WHEN V(out) = VD2
let slope = (lnID2 - lnID1) / (VD2 - VD1)
* let yintercept = lnID1 - slope * VD1
let eta = 1 / ((26/1000)*slope)
print eta                               ; , exp(yintercept)

hardcopy 1b1a.eps I(Vdummy) vs V(out)
hardcopy 1b1b.eps y vs V(out)
.endc
.end
```

3.2.1.3 $I_L = 10mA$

```
Param Rathour (190070049), I/V characteristics of Solar Cell

.include Solar_Cell.txt                  ; Includes Solar Cell Model

R1 mid out 100
X1 out gnd solar_cell IL_val = 10e-3     ; Solar Cell
Vin in gnd dc 0                          ; DC source Vin
Vdummy in mid 0                          ;
.dc Vin -2 2 0.01                        ; DC Analysis
.control
run
plot I(Vdummy) vs V(out)
let y = log({abs(I(Vdummy)+10m)})
plot y vs V(out)
let VD1 = 0
let VD2 = 0.55
meas dc lnID1 FIND y WHEN V(out) = VD1
meas dc lnID2 FIND y WHEN V(out) = VD2
let slope = (lnID2 - lnID1) / (VD2 - VD1)
* let yintercept = lnID1 - slope * VD1
let eta = 1 / ((26/1000)*slope)
print eta                                ; , exp(yintercept)

hardcopy 1b2a.eps I(Vdummy) vs V(out)
hardcopy 1b2b.eps y vs V(out)
.endc
.end
```

3.2.2 Measurement of V_{OC} and I_{SC} and factor of solar cell

3.2.2.1 $I_L = 8mA$

```
Param Rathour (190070049), I/V characteristics of Solar Cell

.include Solar_Cell.txt                  ; Includes Solar Cell Model

R1 in out 0
X1 out gnd solar_cell IL_val = 8e-3      ; Solar Cell
Vdummy in gnd 0                          ;
.dc R1 1 500 0.1                         ; DC Analysis
.control
run
let I_D = -I(Vdummy)
let V_D = V(out)
let Power = I_D*V_D
plot I_D vs V_D
```

```
plot Power vs V_D
let I_SC = minimum(I_D)
let V_OC = maximum(V_D)
let Pmax = minimum(Power)
let FF = Pmax / (I_SC*V_OC)
print I_SC V_OC FF
hardcopy 23a1.eps I_D vs V_D
hardcopy 23a2.eps Power vs V_D
.endc
.end
```

3.2.2.2 $I_L = 10mA$

```
Param Rathour (190070049), I/V characteristics of Solar Cell

.include Solar_Cell.txt                 ; Includes Solar Cell Model

R1 in out 0
X1 out gnd solar_cell IL_val = 10e-3    ; Solar Cell
Vdummy in gnd 0                         ;
.dc R1 1 500 0.1                        ; DC Analysis
.control
run
let I_D = -I(Vdummy)
let V_D = V(out)
let Power = I_D*V_D
plot I_D vs V_D
plot Power vs V_D
let I_SC = minimum(I_D)
let V_OC = maximum(V_D)
let Pmax = minimum(Power)
let FF = Pmax / (I_SC*V_OC)
print I_SC V_OC FF
hardcopy 23b1.eps I_D vs V_D
hardcopy 23b2.eps Power vs V_D
.endc
.end
```

3.2.2.3 C/V characteristics of a Solar Cell

```
Param Rathour (190070049), C-V characteristics of solar_cell

.include Solar_Cell.txt                 ; Includes Solar Cell Model
.include TL071.cir                      ; Includes OpAmp Model

```

```
Vdc dc1 gnd dc 0
Vin in gnd sin(0 0.5 1k 0 0)            ; Source Vin
Vp1 p1 gnd dc 12
Vn1 n1 gnd dc -12
Vp2 p2 gnd dc 12
Vn2 n2 gnd dc -12
R1 in ni1 10k
R2 dc1 ni1 1k
X1 gnd ni1 p1 n1 duti TL071
RF ni1 duti 1k
X2 duti duto solar_cell IL_val = 0      ; Solar Cell
RFB duto out 10k
CFB duto out 4.7n
X3 gnd duto p2 n2 out TL071
.dc Vdc 0 2 0.01                        ; DC Analysis
.control
run
let pi = 3.1415
let k1 = 4.7n*(sqrt(1+1/(2*pi*1k*10k*4.7n)^2))
let Cdut = abs({k1* V(out)/V(duti)})
plot Cdut vs abs(V(dc1))
let k2 = 2/(((8.85e-12)*11.68)*(1.602e-19))
let area = (4e-2)^2
let C = Cdut / area
let y = 1/C^2
plot 1/C^2 vs V(duti)
let VD1 = -1.4
let VD2 = -1.8
meas dc C1 FIND y WHEN V(duti) = VD1
meas dc C2 FIND y WHEN V(duti) = VD2
let slope = (C2 - C1) / (VD2 - VD1)
print slope
let ND = k2/abs(slope)
print ND
let yintercept = C1 - slope*VD1
let Vbi = yintercept*ND/k2
print Vbi
hardcopy 4a.eps Cdut vs abs(V(dc1))
hardcopy 4b.eps 1/C^2 vs V(duti)
.endc
.end
```

3.2.3 Code for Plots

```
set hcopydevtype=postscript
set hcopypscolor = 0
```

```
set color0 = rgb:f/f/f
set color1 = rgb:/1/1/1
set color2 = rgb:f/0/0
set color3 = rgb:0/0/f
set color4 = rgb:0/f/0
```

4 Experiment completion status

Completed everything in Lab but the ideality factor calculation wee wrong as I forgot to consier non zero I_L

5 Questions for reflection

Please see my Design and Simulation Result sections for some answers to questions.
Overall, the lab was good. Again reports took the most of my time.
Thanks!