EE230 - Experiment 4

Param Rathour - 190070049
Autumn Semester 2021-22

1 BJT Current Source (using BC557)

1.1 Analysis

$$V_B = V_{CC} - V_Z = (12 - 5.6) = 6.4V \quad \rightarrow \quad V_E = V_B + V_{EB} = (6.4 + 0.7) = 7.1V$$

$$I_E = \frac{V_{CC} - V_E}{R_E} = \frac{12 - 7.1}{4.7k} = 1.0425mA$$

(i)

$$I_L = I_C = \frac{I_E}{\alpha} = \frac{I_E}{\beta/\beta + 1} = 1.053mA \rightarrow V_L = I_L \cdot R_L \quad \text{(BJT must be in active mode for current source)}$$

For $R_L = 100\Omega \rightarrow V_L = 0.1053V < 6.9V$ & for $R_L = 1k\Omega \rightarrow V_L = 1.053V < 6.9V$, hence our analysis is correct.

(ii)

$$\max R_L = \frac{\max V_L}{I_L} = \frac{V_E + V_{CE_{\text{sat}}}}{I_L} = \frac{7.1 - 0.2}{1.0425} k\Omega = 6.6187k\Omega$$

(iii) So, $R_L = \{100\Omega, 470\Omega, 1k\Omega, 2.2k\Omega, 3.3k\Omega, 4.7k\Omega, 5.6k\Omega\} \rightarrow I_L = 1.0425mA$ and for $R_L 6.8k\Omega, 8.2k\Omega, 10k\Omega, I_L$ is $1.015mA$, $0.8415mA$ and $0.69mA$ respectively.

1.2 NGSPICE Values

$V_L = 1.012825 \cdot 10^0 V, \quad I_L = 1.012825 \cdot 10^{-3} A$. As can be seen, analysis and simulation gives very close values.

1.3 Plots



(a) I_L vs R_L



(b) I_L vs V_L



(c) BJT current source

1.4 Code

```
Param Rathour (190070049), BJT Current Source
.include bc557.txt                                   ; Includes BJT Model
.include zener_B.txt                                 ; Includes Zener 5.6V Model
Q1 c b e bc557a
x1 b Vcc zener_B
VCC Vcc gnd 12                                       ; Supply Voltage
VDL c L 0                                            ; Dummy Voltage
RE Vcc e 4.7k                                        ; Resistor
RB b gnd 2.2k                                        ; Resistor
RL L gnd 1k                                          ; Resistor
.dc RL 1k 10k 1k                                     ; DC Analysis (sweep R_L)
.control                                             ; Control Functions
run
plot I(VDL) vs V(L)
.endc
.end
```

1.5 Learnings

The BJT current source works for wide range of voltage V_L, but after R_L crosses a threshold, current reduces sharply.

2 BJT Current Mirror based Current Source

2.1 Analysis

$$I_{REF} = \frac{V_{CC} - V_{BE}}{R} = \frac{12 - 0.7}{10k\Omega} = 1.13mA$$

$$I_O = \frac{1}{1 + \frac{2}{\beta}} \left(1 + \frac{V_O - V_{BE}}{V_A}\right) \quad \text{(here } \beta = 100\text{)}$$

Using the above two equations,

$$I_{REF} = 1.13345mA \text{ (NGSPICE)}$$

V_O (in V)	V_A (in V)	I_O (in mA) Analysis	I_O (in mA) Simulation
1	80	0.9841	1.10714
5	80	1.0331	1.16402
1	200	0.9819	1.10429
5	200	1.0014	1.12708

Table 2.1: Analysis and Simulation comparison (similar values)

2.2 Plots



(a) $V_A = 80V$



(b) $V_A = 200V$



(c) BJT current mirror

Figure 2.1: I_O vs V_O

2.3 Code

```
Param Rathour (190070049), BJT Current Source based Current Source
.include bc547Va80.txt           ; Includes BJT Model with V_A = 80 (similar file for V_A = 200)
Q1 b1 b1 gnd bc547a
Q2 c2 b1 gnd bc547a
VCC Vcc gnd 12                                     ; Supply Voltage
Vo c2 gnd 1                                        ; Early Voltage V_O = 1V (similar for V_O = 5V)
R Vcc b1 10k                                       ; Resistor
.dc Vo 1 5 0.                                      ; DC Analysis ((sweep V_O))
.control                                           ; Control Functions
run
plot I(Vo)
.endc
.end
```

2.4 Learnings

The variation of Current (I_O) with Voltage (V_O) is linear, as V_O increases I_O magnitude increases
I_O NGSPCIE values only approximately equal with analysis, errors could be due to different BJT model parameters

3 Differential Pair

3.1 DC Operating Values

3.1.1 Analysis

$$I_E = \frac{V_E - (-V_{EE})}{R_E} = \frac{(V_B - V_{BE}) - (-V_{EE})}{R_E} = \frac{-0.7 + 12}{10k} = 1.13mA \rightarrow I_{C_1} = I_{C_2} = \frac{I_E}{2} = 0.565mA \text{ assumming } I_B = 0$$

$$V_E = -0.7V \quad V_{C_1} = V_{C_2} = V_{CC} - I_{C_1} \cdot R_{C_1} = 12 - 0.565 \cdot 10^{-3} \cdot 6.8 \cdot 10^3 = 8.158V$$

3.1.2 NGSPICE Values

$$I_{C_1} = I_{C_2} = 5.630775 \cdot 10^{-4}A \quad I_E = 1.135394 \cdot 10^{-3}A \quad V_E = -6.46059 \cdot 10^{-1}V \quad V_{C_1} = V_{C_2} = 8.171073V$$

As can be seen, the Analysis and Simulation values are fairly accurate

3.1.3 Code

```
Param Rathour (190070049), Differential Pair, DC Operating Values
.include bc547.txt                                     ; Includes BJT Model
Q1 c1 b1 e bc547a
Q2 c2 b2 e bc547a
VCC VCC gnd 12                                         ; Supply Voltage
VEE VEE gnd -12                                        ; Supply Voltage
VC1 V01 c1 0                                           ; Dummy Voltage
VC2 V02 c2 0                                           ; Dummy Voltage
VE e VdE 0                                             ; Dummy Voltage
RC1 VCC V01 6.8k                                       ; Resistor
RC2 VCC V02 6.8k                                       ; Resistor
RB1 b1 gnd 1k                                          ; Resistor
RB2 b2 gnd 1k                                          ; Resistor
RE VdE VEE 10k                                         ; Resistor
.op
.control                                               ; Control Functions
run
print I(VC1) I(VC2) I(VE) V(e) V(C1) V(C2)
.endc
.end
```

3.2 Differential Amplifier

3.2.1 Analysis

$$g_m = \frac{I_C}{V_T} = 0.565/26 = 0.0217\mho \quad \text{Gain} \approx -g_m \cdot R_C/2 = 73.88V/V \quad \text{(Single Ended)}$$

3.2.2 NGSPICE values

$$V_{01_{dbpp}} = -4.881849V, \quad V_{\text{in}_{dbpp}} = -4 \cdot 10^{-1}V, \quad \text{Gain} = 3.511815 \cdot 10^1 db = 57.0043V/V$$

3.2.3 Plots



(a) V_{in_1} in dB (b) V_{O_1} in dB (c) V_{O_2} in dB

3.2.4 Code

```
Param Rathour (190070049), Differential Pair Amplifier

.include bc547.txt                                   ; Includes BJT Model
Q1 c1 b1 e bc547a
Q2 c2 b2 e bc547a
VCC VCC gnd 12                                       ; Supply Voltage
VEE VEE gnd -12                                      ; Supply Voltage
Vin1 Vin1 gnd dc 0 ac 10m                            ; Voltage
Vin2 Vin2 gnd dc 0 ac 0                              ; Voltage
VC1 V01 c1 0                                         ; Dummy Voltage
VC2 V02 c2 0                                         ; Dummy Voltage
VE e VdE 0                                           ; Dummy Voltage
RC1 VCC c1 6.8k                                      ; Resistor
RC2 VCC c2 6.8k                                      ; Resistor
RB1 b1 Vin1 1k                                       ; Resistor
RB2 b2 Vin2 1k                                       ; Resistor
RE VdE VEE 10k                                       ; Resistor
.ac DEC 10 1 500000k                                 ; Transient Analysis
.control                                             ; Control Functions
run
plot Vdb(Vin1) Vdb(V01) Vdb(V02)
meas ac Vdb01pp max vdb(V01)
meas ac Vdbinpp max vdb(Vin1)
let Gain = Vdb01pp - Vdbinpp
print Gain
.endc
.end
```

3.3 Large-Signal Characteristics (Voltage-Transfer Characteristics - VTC)

3.3.1 Plots



(a) V_{C_1} vs V_{in_1} in V



(b) V_{C_2} vs V_{in_1} in V



(c) Differential Pair

3.3.2 Code

Modified code from Section 3.2.4

```
Vin1 Vin1 gnd sin(0 10m 1k 0 0)                      ; Voltage
Vin2 Vin2 gnd sin(0 0 1k 0 0)                        ; Voltage
.dc Vin1 -1 1 0.01                                   ; DC Analysis
plot V(V01) vs V(Vin1)
plot V(V02) vs V(Vin1)
```

3.4 Learnings

I learned about the BJT differential amplifier. It provides high gain which is used to reduces the unwanted signals. The analysis and simulation values are very close except for gain (probably because of approximations in analysis).