EE236: Experiment 2

I-V characteristics of Schottky and Zener Diodes

Param Rathour, 190070049

Spring Semester, 2021-22

Contents

1 Overview of the experiment

This report contains my approach to the experiment, the circuit's design with the relevant simulation code and output plots.

1.1 Aim of the experiment

- To plot and understand I/V characteristics of different diodes and compare their parameters from the plot
- To design the circuits for given transfer characteristics
- To simulate Zener Regulator and Voltage Doubler

1.2 Methods

First a comparison between Schottky BAT85, Schottky BAT 960 and Diode 1N914 was made using the parameters from their I/V characteristics. The parameters were Cut-in Voltages and Reverse Recovery Time (t_{rr}).
Then, Zener Diode's I/V forward and reverse characteristics was plotted. This zener diode was used Voltage Regulator experiment for the input voltage as unregulate power supply (bridge rectifier output with capacitor).
Then, Shunt Clippers were used to design circuits from their Transfer Characteristics.
Finally, Voltage Doubler circuit was simulated and observed.

2 Design

I/V characteristics circuits were taken from Experiment 1 Handout. Voltage was varied from -5 to $+5$ in case of Zener Diode. To get t_{rr}, a pulse of frequency 100kHz with amplitudes $-5, 5$ was used as input in the below circuit.



(a) For Schottky and 1N914

(b) For Zener

Figure 1: Diode Characteristics Circuit

Schottky Diode were used in the Rectifier Circuits (without the capacitor) given below to get rectified outputs.
These outputs were then used as inputs to Voltage Regulator.
In Voltage Regulator, Zener Diode maintains a steady voltage of 5.6V when $R_S \geq 52.64\Omega$,
As max $1W$ power can flow in R_S, the maximum possible current flowing through it will be $\frac{1}{5.6}$ when as breakdown voltage $= 5.6V$.

This gives, $V_Z = 5.6V \rightarrow V_S = V_Z + R_S \cdot I_S$, $R_S = \dfrac{(15-5.6)}{\frac{1}{5.6}} = 52.64$

The negative terminal of opamp is also $5.6V$ due to Virtual Short.
So, $R_1 = 3.4k\Omega$ and $R_2 = 5.6k\Omega$ satisfy this, as the output required is $9V$.
The BJT amplifies the current in R_S as well as provides feedback mechanism to regulate this current.



(a) Unregulated Power Supply Generation



(b) Voltage Regulator

Figure 2: Zener Diode in Voltage Regulator Experiment

With the idea of Shunt Clippers and a Battery to shift the limit, the following circuits were designed to generate given transfer characteristics



(a)



(b)

Figure 3: Designing circuit from transfer characteristics

The given Voltage Doubler was used without any modifications
I have taken 20V sine as input to system.



Figure 4: Voltage Doubler

3 Simulation results

3.1 Plots



(a) 1N914



(b) Schottky BAT85



(c) Schottky BAT960



(d) Zener

Figure 5: I/V Characteristics of Different Diodes

Diode 1N914 has cut-in voltage $0.48V$ whereas for Schottky diodes it is 0.163 and $0.107V$ respectively.
Zener Diode has breakdown voltage in excess of $5V$



(a) 1N914 (b) Schottky BAT85 (c) Schottky BAT960



(d) 1N914 (e) Schottky BAT85 (f) Schottky BAT960

Figure 6: Rectifier Circuit (above V_{out} and V_{in}, below V_{out} vs V_{in})

Rectifed output is lesser amplitude in Diode 1N914 than Schottky diodes due to the differences in cut-in voltages.



(a) 1N0007 (b) Schottky BAT85 (c) Schottky BAT960

Figure 7: Reverse Recovery Time

Diode 1N4007 has Reverse Recovery Time $\approx 1795ns$ whereas for Schottky diodes it is $\approx 6.17ns$ and $\approx 39.487ns$ respectively. So Schottky diodes can be used in applications where fast swithching is required like frequency mixer and detector, power rectifer



(a) V_{out} vs V_{in}

(b) V_{out}

Figure 8: Voltage Regulator

The output voltage is nicely regulated around $9V$



(a) For Circuit in Figure 3a

(b) For Circuit in Figure 3b

Figure 9: Designing circuit from transfer characteristics

The transfer characteristics are very much like given in the handout.



(a) Output Voltage (also Voltage across C_2)

(b) Voltage across C_1

Figure 10: Voltage Doubler

The voltages across the capacitors C_1 and C_2 is approximately $9V$ and $17.8V$.

3.2 Code Snippets

3.2.1 I/V Characteristics

3.2.1.1 Diode 1N914

```
Param Rathour (190070049), I/V characteristics of Diode 1N914

.include Diode_1N914.txt                  ; Includes Diode Model
R1 in mid1 100                            ; Resistor
R2 mid1 gnd 1k                            ; Resistor
R3 mid3 gnd 100                           ; Resistor
dR mid1 midR 1N914                        ; Diode
VdR midR mid3 0                           ; Dummy Voltage to measure I_d
Vin in gnd dc 0                           ; DC source Vin
.dc Vin 0 5 0.01                          ; DC Analysis
.control
run
plot I(VdR) vs {V(mid1) - V(mid3)}
let I1 = 100u
meas dc Vm1I1 FIND V(mid1)  WHEN I(VdR) = I1
meas dc Vm3I1 FIND V(mid3)  WHEN I(VdR) = I1
print Vm1I1-Vm3I1
.endc
```

```
.end
```

3.2.1.2 Schottky BAT85

```
Param Rathour (190070049), I/V characteristics of Schottky BAT85

.include schottky_BAT85.txt              ; Includes Diode Model
R1 in mid1 100                           ; Resistor
R2 mid1 gnd 1k                           ; Resistor
R3 mid3 gnd 100                          ; Resistor
xR mid1 midR BAT85                       ; Diode
VdR midR mid3 0                          ; Dummy Voltage to measure I_d
Vin in gnd dc 0                          ; DC source Vin
.dc Vin 0 5 0.01                         ; DC Analysis
.control
run
plot I(VdR) vs {V(mid1) - V(mid3)}
let I1 = 100u
meas dc Vm1I1 FIND V(mid1)  WHEN I(VdR) = I1
meas dc Vm3I1 FIND V(mid3)  WHEN I(VdR) = I1
print Vm1I1-Vm3I1
.endc
.end
```

3.2.1.3 Schottky BAT960

```
Param Rathour (190070049), I/V characteristics of Schottky BAT960

.include schottky_BAT960.txt             ; Includes Diode Model
R1 in mid1 100                           ; Resistor
R2 mid1 gnd 1k                           ; Resistor
R3 mid3 gnd 100                          ; Resistor
xR mid1 midR BAT960                      ; Diode
VdR midR mid3 0                          ; Dummy Voltage to measure I_d
Vin in gnd dc 0                          ; DC source Vin
.dc Vin 0 5 0.01                         ; DC Analysis
.control
run
plot I(VdR) vs {V(mid1) - V(mid3)}
let I1 = 100u
meas dc Vm1I1 FIND V(mid1)  WHEN I(VdR) = I1
meas dc Vm3I1 FIND V(mid3)  WHEN I(VdR) = I1
print Vm1I1-Vm3I1
.endc
.end
```

3.2.1.4 Zener Diode

```
Param Rathour (190070049), I/V characteristics of Zener Diode

.include zener.txt                          ; Includes Diode Model
R1 in mid1 100                              ; Resistor
R2 mid1 gnd 1k                              ; Resistor
R3 mid3 gnd 100                             ; Resistor
xR mid1 midR DI_1N4734A                     ; Diode
VdR midR mid3 0                             ; Dummy Voltage to measure I_d
Vin in gnd dc 0                             ; DC source Vin
.dc Vin -10 5 0.01                          ; DC Analysis
.control
run
plot I(VdR) vs {V(mid1) - V(mid3)}
hardcopy 1d1.eps I(VdR) vs {V(mid1) - V(mid3)}
.endc
.end
```

3.2.2 Diode-Based Bridge Rectifier

3.2.2.1 Diode 1N914

```
Param Rathour (190070049), Diode-Based Bridge Rectifier

.include Diode_1N914.txt                    ; Includes Diode Model
Vin in gnd sin(0 16.9705627 50 0 0)         ; Source Vin

D1 in outp 1N914                            ; Diode
D2 outn in 1N914                            ; Diode
D3 outn gnd 1N914                           ; Diode
D4 gnd outp 1N914                           ; Diode
RL outp outn 1k                             ; Resistor (assumed 1k)

.tran 0.01m 0.12                            ; Transient Analysis
.control
run
plot V(in) {V(outp) - V(outn)}
plot {V(outp) - V(outn)} vs V(in)

.endc
.end
```

3.2.2.2 Schottky BAT85

```
Param Rathour (190070049), Diode-Based Bridge Rectifier

.include schottky_BAT85.txt                ; Includes Diode Model
Vin in gnd sin(0 16.9705627 50 0 0)       ; Source Vin

x1 in outp BAT85                          ; Diode
x2 outn in BAT85                          ; Diode
x3 outn gnd BAT85                         ; Diode
x4 gnd outp BAT85                         ; Diode
RL outp outn 1k                           ; Resistor (assumed 1k)

.tran 0.01m 0.12                          ; Transient Analysis
.control
run
plot V(in) {V(outp) - V(outn)}
plot {V(outp) - V(outn)} vs V(in)

.endc
.end
```

3.2.2.3 Schottky BAT960

```
Param Rathour (190070049), Diode-Based Bridge Rectifier

.include schottky_BAT960.txt               ; Includes Diode Model
Vin in gnd sin(0 16.9705627 50 0 0)       ; Source Vin

x1 in outp BAT960                         ; Diode
x2 outn in BAT960                         ; Diode
x3 outn gnd BAT960                        ; Diode
x4 gnd outp BAT960                        ; Diode
RL outp outn 1k                           ; Resistor (assumed 1k)

.tran 0.01m 0.12                          ; Transient Analysis
.control
run
plot V(in) {V(outp) - V(outn)}
plot {V(outp) - V(outn)} vs V(in)

.endc
.end
```

3.2.3 Reverse Recovery Time

3.2.3.1 1N4007

```
Param Rathour (190070049), Reverse Recovery Time of Diode 1N4007

.include 1N4007.txt                              ; Includes Diode Model
R1 in mid1 100                                   ; Resistor
R2 mid1 gnd 1k                                   ; Resistor
R3 mid3 gnd 100                                  ; Resistor
dR mid1 midR DI_1N4007                           ; Diode
VdR midR mid3 0                                  ; Dummy Voltage to measure I_d
Vin in gnd pulse(-5 5 0 0 0 0.005m 0.01m)        ; Pulsed input Vin
.tran 0.001u 7u 4.99u                            ; Transient Analysis
.control
run
plot I(VdR)                                      ; vs {V(mid1) - V(mid3)}
hardcopy 1c0.eps I(VdR)
meas trans Imin MIN I(Vdr)
meas trans t2 WHEN I(Vdr) = -1e-5 RISE = 1
meas trans t1 WHEN I(Vdr) = Imin
print t2 - t1
.endc
.end
```

3.2.3.2 Schottky BAT85

```
Param Rathour (190070049), Reverse Recovery Time of Schottky BAT85

.include schottky_BAT85.txt                      ; Includes Diode Model
R1 in mid1 100                                   ; Resistor
R2 mid1 gnd 1k                                   ; Resistor
R3 mid3 gnd 100                                  ; Resistor
xR mid1 midR BAT85                               ; Diode
VdR midR mid3 0                                  ; Dummy Voltage to measure I_d
Vin in gnd pulse(-5 5 0 0 0 0.005m 0.01m)        ; Pulsed input Vin
.tran 0.001u 5.1u 4.99u                          ; Transient Analysis
.control
run
plot I(VdR)                                      ; vs {V(mid1) - V(mid3)}
hardcopy 1c0.eps I(VdR)
meas trans Imin MIN I(Vdr)
meas trans t2 WHEN I(Vdr) = -1e-5 RISE = 1
meas trans t1 WHEN I(Vdr) = Imin
print t2 - t1
.endc
```

```
.end
```

3.2.3.3 Schottky BAT960

```
Param Rathour (190070049), Reverse Recovery Time of Schottky BAT960

.include schottky_BAT960.txt              ; Includes Diode Model
R1 in mid1 100                            ; Resistor
R2 mid1 gnd 1k                            ; Resistor
R3 mid3 gnd 100                           ; Resistor
xR mid1 midR BAT960                       ; Diode
VdR midR mid3 0                           ; Dummy Voltage to measure I_d
Vin in gnd pulse(-5 5 0 0 0 0.005m 0.01m) ; Pulsed input Vin
.tran 0.001u 5.1u 4.99u                   ; Transient Analysis
.control
run
plot I(VdR)                               ; vs {V(mid1) - V(mid3)}
hardcopy 1c0.eps I(VdR)
meas trans Imin MIN I(Vdr)
meas trans t2 WHEN I(Vdr) = -1e-5 RISE = 1
meas trans t1 WHEN I(Vdr) = Imin
print t2 - t1
.endc
.end
```

3.2.4 Voltage Regulator

```
Param Rathour (190070049), Zener Diode in Voltage Regulator Experiment

.include zener.txt                        ; Includes Diode Model
.include ua741.txt                        ; Includes OpAmp Model
.include bc547.txt                        ; Includes BJT Model
.include Diode_1N914.txt                  ; Includes Diode Model

Vin in gnd sin(0 16.7 50 0 0)             ; Source Vin
Vop op gnd dc 12                          ; DC Source Vin
Von on gnd dc -12                         ; DC Source Vin

D1 in outp 1N914                          ; Diode
D2 outn in 1N914                          ; Diode
D3 outn gnd 1N914                         ; Diode
D4 gnd outp 1N914                         ; Diode
RL outp outn 1k                           ; Resistor (assumed 1k)
CL outp outn 100u                         ; Capacitor

q1 outp b e bc547a
```

```
x1 mid1 mid2 op on b UA741
RS outp mid1 100                          ; Resistor
R1 e mid2 3.4k                            ; Resistor
R2 mid2 outn 5.6k                         ; Resistor
xR outn mid1  DI_1N4734A                  ; Diode
.tran 0.1m 100m                           ; Transient Analysis
.control
run
plot {V(e) - V(outn)} vs {V(outp) - V(outn)}
plot {V(e) - V(outn)}                     ; {V(outp) - V(outn)}
* plot I(VdR) vs {V(mid1) - V(mid3)}
hardcopy 1d21.eps {V(e) - V(outn)} vs {V(outp) - V(outn)}
hardcopy 1d22.eps {V(e) - V(outn)}
.endc
.end
```

3.2.5 Designing circuit from transfer characteristics

3.2.5.1 Circuit 3a

```
Param Rathour (190070049), Shunt Clipper custom circuit 1

.include Diode_1N914.txt                  ; Includes Diode Model
R1 out inp 1k                             ; Resistor (assumed 1k)
D1 out mid 1N914                          ; Diode
Vdc mid gnd dc 2                          ; Independent DC source
Vin inp inn sin(0 5 1k 0 0)               ; AC source Vin
Vc inn gnd dc 0.7
.tran 0.01m 6m                            ; Transient Analysis
.control
run
* plot {V(inp)-V(inn)} V(out)
plot V(out) vs {V(inp)-V(inn)}
hardcopy 1e1.eps V(out) vs {V(inp)-V(inn)}
.endc
.end
```

3.2.5.2 Circuit 3b

```
Param Rathour (190070049), Shunt Clipper custom circuit 2

.include Diode_1N914.txt                  ; Includes Diode Model
R1 out in 1k                              ; Resistor (assumed 1k)
D1 out mid1 1N914                         ; Diode
D2 mid2 out 1N914                         ; Diode
Vdc1 mid1 gnd dc 5                        ; Independent DC source
```

```
Vdc2 mid2 gnd dc -4.2                       ; Independent DC source
Vin in gnd sin(0 8 1k 0 0)                  ; AC source Vin
.tran 0.01m 6m                              ; Transient Analysis
.control
run
* plot {V(inp)-V(inn)} V(out)
plot V(out) vs V(in)
hardcopy 1e2.eps V(out) vs V(in)
.endc
.end
```

3.2.6 Voltage Doubler

```
Param Rathour (190070049), Simple application circuit using diode and capacitor: Voltag

.include Diode_1N914.txt                    ; Includes Diode Model
R1 out gnd 4.7k                             ; Resistor (assumed 1k)
C1 mid in 10u                               ; Capacitor
C2 out gnd 10u                              ; Capacitor
D1 gnd mid 1N914                            ; Diode
D2 mid out 1N914                            ; Diode
Vin in gnd sin(0 10 1k 0 0)                 ; AC source Vin
.tran 0.01m 10m                             ; Transient Analysis
.control
run
plot V(out)
plot {V(mid)-V(in)}

hardcopy 1f1.eps V(out)
hardcopy 1f2.eps {V(mid)-V(in)}
.endc
.end
```

4 Experiment completion status

I was down with viral fever and hence could not attempt this lab during lab session. I attempted it as I felt better. My assigned TA will conduct in-lab demo with new timings.

5 Questions for reflection

Please see my Design and Simulation Result sections for some answers to questions.

- Zener diode behaviour is different in the reverse bias region due to Zener Breakdown which is different than Avalanche breakdown. It can be used as Voltage Regulator.

- The BJT amplifies the current in R_S as well as provides feedback mechanism to regulate this current.
- The factor that limits the load of regulator is the current by BJT (gain from opamp). To overcome this another amplifier can be used.
- The output voltage of doubler circuit fluctuates around $17.8V$. The voltages across the capacitors C_1 and C_2 is approximately $9V$ and $17.8V$. If a smaller resistance ($4.7k\Omega$) is used then capacitors discharges faster due to less time constant. This results in ripples in output voltage.

Overall, the lab was good. Again reports took the most of my time.
Thanks!