EE230 - Experiment 10

Param Rathour - 190070049
Autumn Semester 2021-22

1 Wien-bridge Oscillator

1.1 Circuit & Plots



(a) Basic Wien-bridge oscillator



(b) V_{out} vs time for 1a



(c) Wien-bridge oscillator with amplitude stabilization



(d) V_{out} vs time for 1c

1.2 Code

```
Param Rathour (190070049), Wien-bridge oscillator

.include IN914.txt                                                   : Includes Diode Model
.include ua741.txt                                                   : Includes Op-amp Model

VCCp VCCp gnd 12                                                     : Supply Voltage
VCCn VCCn gnd -12                                                    : Supply Voltage
x1 Vp Vn VCCp VCCn Vo ua741                                          : Operational Amplifier

R1 gnd Vn 4.7k                                                       : Resistor
R2 Vn Vo 10k                                                         : Resistor
RF1 Vo Vmid 4.7k                                                     : Resistor
CF1 Vmid Vp 0.1u                                                     : Capacitor
RF2 Vp gnd 4.7k                                                      : Resistor
CF2 Vp gnd 0.1u                                                      : Capacitor

.tran 0.01m 525m 500m                                                : Transient Analysis
.control                                                             : Control Functions
run
meas trans Vpeak max V(Vo)
plot V(Vo)

.endc
.end
```

Remove R_2 and add the follwing lines to simulate Basic Wien-bridge oscillator with amplitude stabilization

```
R2A Vn Vd 6.8k                                                       : Resistor
R2B Vd Vo 3.3k                                                       : Resistor
D1 Vo Vd 1N914                                                       : Diode
D2 Vd Vo 1N914                                                       : Diode
```

1.3 Learnings

V_{out} gets saturated in Basic Wien-bridge. By adding signal diodes, we introduce non-linearity and prevent saturation. This reduces the gain as V_{out} increases resulting in pure sine of amplitude $\approx 2.688185V$ (well below saturation ($12V$)).

2 Phase-shift Oscillator

2.1 Circuit & Plots



(a) Phase-shift Oscillator



(b) V_{out} vs time

2.2 Code

```
Param Rathour (190070049), Phase-shift Oscillator

.include IN914.txt                                        : Includes Diode Model
.include ua741.txt                                        : Includes Op-amp Model

VCCp VCCp gnd 12                                          : Supply Voltage
VCCn VCCn gnd -12                                         : Supply Voltage
x1 gnd Vn VCCp VCCn Vo ua741                              : Operational Amplifier

C1 Vn Vm1 0.022u                                          : Capacitor
R1 Vm1 gnd 10k                                            : Resistor
C2 Vm1 Vm2 0.022u                                         : Capacitor
R2 Vm2 gnd 10k                                            : Resistor
C3 Vm2 Vo 0.022u                                          : Capacitor
RF Vn Vo 200k                                             : Resistor

.tran 0.01m 525m 500m                                     : Transient Analysis
.control                                                  : Control Functions
run
meas trans Vpeak max V(Vo)
plot V(Vo)

.endc
.end
```

2.3 Learnings

Here too, we get clipping and amplitude stabilization is necessary.

3 Astable Multivibrator

3.1 Circuit & Plots



(a) V_C, V_{out} vs time



(b) V_C, V_{out} vs time



(c) V_C, V_{out} vs time



(d) Astable Multivibrator

3.2 Values

Parameter	V_{IH}	V_{IL}	V_{OH}	V_{OL}
$C = 0.1\mu F$	$3.170581V$	$-3.17326V$	$6.190714V$	$-6.19071V$
$C = 0.01\mu F$	$3.575057V$	$-3.587469V$	$6.190759V$	$-6.190749V$
$C = 0.001\mu F$	$4.352380V$	$-4.360387V$	$6.087647V$	$-6.08480V$

Slew Rate in case c) is $\approx 0.452132V/\mu s \left(= \frac{4.78261-(-0.815217)}{0.500047-0.50006}\right)$. This is very close to value in datasheet $0.5V/\mu s$.

3.3 Code

```
Param Rathour (190070049), Astable Multivibrator

.include zener_B.txt
.include ua741.txt                                      : Includes Op-amp Model

VCCp VCCp gnd 12                                        : Supply Voltage
VCCn VCCn gnd -12                                       : Supply Voltage
x1 Vp Vn VCCp VCCn Vo1 ua741                            : Operational Amplifier
x2 Vo Vm zener_B
x3 gnd Vm zener_B
RG Vn Vo 10k                                            : Resistor
CG Vn gnd 0.1u                                          : Capacitor
R3 Vo1 Vo 100                                           : Resistor
R1 Vp Vo 10k                                            : Resistor
R2 Vp gnd 10k                                           : Resistor

.tran 0.01m 525m 500m                                   : Transient Analysis
.control                                                : Control Functions
run
meas trans VIH max V(Vn)
meas trans VIL min V(Vn)
meas trans VOH max V(Vo)
meas trans VOL min V(Vo)
plot V(Vn) V(Vo)
print VIH VIL VOH VOL
.endc
.end
```

R_3 was taken as 100Ω.

3.4 Learnings

Input and output level voltages are reasonable.
But as the capacitance decreases, the performance worsens due to slew rate limitations.