EE230 - Experiment 8

Param Rathour - 190070049
Autumn Semester 2021-22

1 Improved Half Wave Rectifier A

1.1 Circuit & Plots



(a) Circuit



(b) $V_{\text{in}}, V_{\text{out}}$ vs t



(c) V_{out} vs V_{in}

1.2 Code

```
Param Rathour (190070049), Improved Half Wave Rectifier A

.include IN914.txt                                  ; Includes Diode Model
.include ua741.txt                                  ; Includes Op-amp Model

Vi Vi gnd sin(0 1 100 0 0)                          ; Input Voltage
VCCp VCCp gnd 12                                    ; Supply Voltage
VCCn VCCn gnd -12                                   ; Supply Voltage
x1 gnd Vn VCCp VCCn Vo1 ua741                       ; Operational Amplifier
d1 Vn Vo1 IN914                                     ; Diode
d2 Vo1 Vo IN914                                     ; Diode

R1 Vi Vn 10k                                        ; Resistor
R2 Vn Vo 10k                                        ; Resistor
RL Vo gnd 1k                                        ; Resistor

.tran 0.1m 30m                                      ; Transient Analysis
.control                                            ; Control Functions
run
plot V(Vi) V(Vo)
plot V(Vo) vs V(Vi)
.endc
.end
```

1.3 Learnings

When $V_i > 0$, D_1 conducts and D_2 is off, whereas D_2 conducts and D_1 is off when $V_i < 0$
This rectifier operates in the linear region for both positive and negative values of V_i. Hence, better high-frequency performance compared to precision rectifier circuit.
High-frequency performance still isn't much good due to Op-amp's (UA741) poor performance.

2 Full Wave Rectifier

2.1 Circuit & Plots



(a) $V_{\text{in}}, V_{\text{out}}$ vs t at $f = 100\text{Hz}$



(b) V_{out} vs V_{in} at $f = 100\text{Hz}$



(c) $V_{\text{in}}, V_{\text{out}}$ vs t at $f = 5k\text{Hz}$



(d) V_{out} vs V_{in} at $f = 5k\text{Hz}$



(e) Circuit

2.2 Code

```
Param Rathour (190070049), Full Wave Rectifier

.include IN914.txt                                       ; Includes Diode Model
.include ua741.txt                                       ; Includes Op-amp Model

* Vi Vi gnd dc 0 ac 1                                    ; Input Voltage
Vi Vi gnd sin(0 1 5000 0 0)                              ; Input Voltage
VCCp VCCp gnd 12                                         ; Supply Voltage
VCCn VCCn gnd -12                                        ; Supply Voltage
x1 gnd Vn1 VCCp VCCn Vo1 ua741                           ; Operational Amplifier
x2 gnd Vn2 VCCp VCCn Vo ua741                            ; Operational Amplifier
d1 Vo1 Vn1 1N914                                         ; Diode
d2 Vo2 Vo1 1N914                                         ; Diode
R1 Vi Vn1 10k                                            ; Resistor
R2 Vn1 Vo2 10k                                           ; Resistor
R3 Vi Vn2 10k                                            ; Resistor
R4 Vo2 Vn2 5k                                            ; Resistor
R5 Vn2 Vo 5k                                             ; Resistor
RL Vo gnd 1k                                             ; Resistor
.tran 0.1m 30m                                           ; Transient Analysis (100Hz)
* .tran 0.02m 0.6m                                       ; Transient Analysis (5kHz)
.control                                                 ; Control Functions
run
plot V(Vi) V(Vo)
plot V(Vo) vs V(Vi)
.endc
.end
```

2.3 Learnings

Here, V_o is equal to $|V_i|$. The multiplication of -2 to V_{o_2} and -1 to V_i ensures that their summing of gives $|V_i|$ High-frequency performance still isn't much good due to Op-amp's (UA741) poor performance.

3 Single-pole Active Low-pass Filter

3.1 Analysis

Theoretical Values

$$f_c = \frac{1}{2\pi R_A C_A} = \frac{1}{2\pi} \approx 338.63Hz \quad \text{Roll-off Frequency} = -20\text{dB/decade}$$

Simulation Values

$$f_c = f \text{ when } V_{\text{out}} = \frac{V_{\max}}{\sqrt{2}} \Rightarrow V_{\text{out}_{\text{dB}}} = V_{\max_{\text{dB}}} - 3 \Rightarrow f_c = 3.378190{\cdot}10^2 Hz \quad \text{Roll-off Frequency} = -1.95728{\cdot}10^1 \text{dB/decade}$$

3.2 Circuit & Plots





(b) Circuit

(a) Filter Response for $f = 1$ to $100k$ Hz

3.3 Code

```
Param Rathour (190070049), Single-pole Active Low-pass Filter
.include ua741.txt                                   ; Includes Op-amp Model
Vi Vi gnd dc 0 ac 1m                                 ; Input Voltage
VCCp VCCp gnd 12                                     ; Supply Voltage
VCCn VCCn gnd -12                                    ; Supply Voltage
x1 Vp Vn VCCp VCCn Vo ua741                          ; Operational Amplifier
RA Vi Vp 4.7k                                        ; Resistor
CA Vp gnd 0.1u                                       ; Capacitor
R1 Vn Vo 9.1k                                        ; Resistor
R2 Vn gnd 1k                                         ; Resistor
.ac DEC 100 1 100k                                   ; AC Analysis
.control                                             ; Control Functions
run
meas ac Vdbmax max vdb(Vo)
let Vdbreq = Vdbmax-3
meas ac fC when vdb(Vo) = Vdbreq fall = 1
meas ac V1 find Vdb(Vo) at = 1000
meas ac V2 find Vdb(Vo) at = 10000
let RollOff =  V2 - V1
print fC RollOff
.endc
.end
```

3.4 Learnings

The theoretical results match well with simulation results for cutoff and roll-off frequency.

4 Sallen-Key (2-pole) Active High-pass Filter

4.1 Analysis

Theoretical Values

$$f_c = \frac{1}{2\pi R_A C_A} = \frac{1}{2\pi} \approx 338.63Hz \quad \text{Roll-off Frequency} = 40\text{dB/decade}$$

Simulation Values

$$f_c = f \text{ when } V_{\text{out}} = \frac{V_{\max}}{\sqrt{2}} \Rightarrow V_{\text{out}_{\text{dB}}} = V_{\max_{\text{dB}}} - 3 \Rightarrow f_c = 3.487332{\cdot}10^2 Hz \quad \text{Roll-off Frequency} = 3.992473{\cdot}10^1 \text{dB/decade}$$

4.2 Circuit & Plots



(a) Filter Response for $f = 1$ to $100k$ Hz



(b) Circuit

4.3 Code

```
Param Rathour (190070049), Sallen-Key (2-pole) Active High-pass Filter
.include ua741.txt                                   ; Includes Op-amp Model
Vi Vi gnd dc 0 ac 1m                                 ; Input Voltage
VCCp VCCp gnd 12                                     ; Supply Voltage
VCCn VCCn gnd -12                                    ; Supply Voltage
x1 Vp Vn VCCp VCCn Vo ua741                          ; Operational Amplifier
RA mid Vo 4.7k                                       ; Resistor
CA Vi mid 0.1u                                       ; Capacitor
RB Vp gnd 4.7k                                       ; Resistor
CB mid Vp 0.1u                                       ; Capacitor
R1 Vn Vo 9.1k                                        ; Resistor
R2 Vn gnd 1k                                         ; Resistor
.ac DEC 100 1 100k                                   ; AC Analysis
.control                                             ; Control Functions
run
plot Vdb(Vo)
meas ac Vdbmax max vdb(Vo)
let Vdbreq = Vdbmax-3
meas ac fC when vdb(Vo) = Vdbreq rise = 1
meas ac V1 find Vdb(Vo) at = 10
meas ac V2 find Vdb(Vo) at = 100
let RollOff =  V2 - V1
print fC RollOff
.endc
.end
```

4.4 Learnings

The theoretical results match well with simulation results for cutoff and roll-off frequency.
For very high frequencies, output will again decrease due to Op-amp's (UA741) poor performance.